
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of idle mode test case 6.1.2.7 which are part of the LTE test suite. 

The test case can be demonstrated to run with LTE UEs (see section 5). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Cases:
tc_6_1_2_7
Test Group:
Idle mode
ATS Version:
iwd-EUTRA-B2009-12_D10wk33
System Simulator used:
Anite Conformance Toolset
UE(s)  used:
Huawei CP V100
Verification Status:
PASS
4 Corrections required for test case 6.1.2.7
4.1 Introduction

This section describes the changes required to make test case 6.1.2.7 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-B2009-12_D10wk33 release.

4.2 Change 1

	Testcase name
	f_TC_6_1_2_7_EUTRA

	Reason for change
	1. SIB5 for cell 1 is including frequencies of 3 cells by default. This is no longer required as per R5-104169.
2. SIB5 not configured for cells 12, cell 13

	Summary of change
	1. SIB5 for cell 1 is modified to include frequencies of cell 12, cell 13.
2. SIB5 is configured for cells 12, cell 13 by calling new function f_EUTRA_CellInfo_InitSIB5_6_1_2_7


	Source of change
	Idle_CellReselection.ttcn, EUTRA_CellInfo.ttcn, EUTRA_Component.ttcn


Before:

	  function f_TC_6_1_2_7_EUTRA ( ) runs on EUTRA_PTC {

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template (value) CellPowerList_Type v_CellPowerList_Initial := {

      cs_CellPower ( eutra_Cell1, -67 ),

      cs_CellPower ( eutra_Cell12, -82 ),

      cs_CellPower ( eutra_Cell13, -97 )

    };

    var CellBarred_Type v_CellBarred := barred;

    var CarrierFreqEUTRA v_CarrierFreq_Cell1, v_CarrierFreq_Cell12, v_CarrierFreq_Cell13;

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency;

    var Ul_Bandwidth_Type v_UL_Bandwidth := f_ConvertDL_BandwidthToUL ( px_ePrimaryBandChannelBandwidth) ;

    var Dl_Bandwidth_Type v_DL_Bandwidth := px_ePrimaryBandChannelBandwidth;

    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer;

    //Initialise all cell, security and mobile parameters

    //SIB5 is init. in f_EUTRA_InitSystemInformation5 with default values according to 36.508 cl. 6.3.1.2

    f_EUTRA_Init ( c3 );

    //Get DL carrier frequency of cell 1

    v_CarrierFreq_Cell1 := f_EUTRA_CellInfo_GetEARFCN ( eutra_Cell1 );

    v_CarrierFreq_Cell12 := f_EUTRA_CellInfo_GetEARFCN ( eutra_Cell12 );

    v_CarrierFreq_Cell13 := f_EUTRA_CellInfo_GetEARFCN ( eutra_Cell13 );

    //Get band dependent parameters

    v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam ( v_DL_Bandwidth, v_UL_Bandwidth );

    //Configure SIB5 for Cell 12 according to Table 6.1.2.7.3.3-6

    //CellReselectionPriority for Cell 1 not present

    f_EUTRA_CellInfo_SetSIB5 ( eutra_Cell12, cs_508_SystemInformationBlockType5_Def ( { cs_InterFreqCarrierFreq_CRPri ( v_CarrierFreq_Cell1,

                                                                                         v_ChannelBandwidthDependency.AllowedMeasBandwidth,

                                                                                         omit ),

                                                                                       cs_InterFreqCarrierFreq ( v_CarrierFreq_Cell13,

                                                                                         v_ChannelBandwidthDependency.AllowedMeasBandwidth ) } ) );

    //Configure SIB5 for Cell 13 according to Table 6.1.2.7.3.3-7

    //CellReselectionPriority for Cell 1 not present

    f_EUTRA_CellInfo_SetSIB5 ( eutra_Cell13, cs_508_SystemInformationBlockType5_Def ( { cs_InterFreqCarrierFreq_CRPri ( v_CarrierFreq_Cell1,

                                                                                         v_ChannelBandwidthDependency.AllowedMeasBandwidth,

                                                                                         omit ),

                                                                                       cs_InterFreqCarrierFreq ( v_CarrierFreq_Cell12,

                                                                                         v_ChannelBandwidthDependency.AllowedMeasBandwidth ) } ) );

    //Set maximum cell powel level for Cell 1 & 12 to be used in creation

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell1, -67 );

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell12, -82 );




After:

	     function f_TC_6_1_2_7_EUTRA ( ) runs on EUTRA_PTC {

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template (value) CellPowerList_Type v_CellPowerList_Initial := {

      cs_CellPower ( eutra_Cell1, -67 ),

      cs_CellPower ( eutra_Cell12, -82 ),

      cs_CellPower ( eutra_Cell13, -97 )

    };

    var CellBarred_Type v_CellBarred := barred;

    var CarrierFreqEUTRA v_CarrierFreq_Cell1, v_CarrierFreq_Cell12, v_CarrierFreq_Cell13;

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency;

    var Ul_Bandwidth_Type v_UL_Bandwidth := f_ConvertDL_BandwidthToUL ( px_ePrimaryBandChannelBandwidth) ;

    var Dl_Bandwidth_Type v_DL_Bandwidth := px_ePrimaryBandChannelBandwidth;

    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer;

    var template (value) EUTRA_CellInfo_Type v_CellInfo12, v_CellInfo13;
    //Initialise all cell, security and mobile parameters

    //SIB5 is init. in f_EUTRA_InitSystemInformation5 with default values according to 36.508 cl. 6.3.1.2

    f_EUTRA_Init ( c3 );

    //Get DL carrier frequency of cell 1

    v_CarrierFreq_Cell1 := f_EUTRA_CellInfo_GetEARFCN ( eutra_Cell1 );

    v_CarrierFreq_Cell12 := f_EUTRA_CellInfo_GetEARFCN ( eutra_Cell12 );

    v_CarrierFreq_Cell13 := f_EUTRA_CellInfo_GetEARFCN ( eutra_Cell13 );

    //Get band dependent parameters

    v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam ( v_DL_Bandwidth, v_UL_Bandwidth );

    f_EUTRA_CellInfo_SetSIB5_6_1_2_7 ( eutra_Cell1, cs_508_SystemInformationBlockType5_Def ( { cs_InterFreqCarrierFreq ( v_CarrierFreq_Cell12,

                                                                                         v_ChannelBandwidthDependency.AllowedMeasBandwidth),

                                                                                       cs_InterFreqCarrierFreq ( v_CarrierFreq_Cell13,

                                                                                         v_ChannelBandwidthDependency.AllowedMeasBandwidth ) } ) );

    v_CellInfo12 := f_EUTRA_CellInfo_InitSIB5_6_1_2_7 (eutra_Cell12,

                                                     px_ePrimaryFrequencyBand,

                                                     c3,

                                                     px_ePrimaryBandChannelBandwidth,

                                                     v_UL_Bandwidth);

    f_EUTRA_CellInfo_Set (eutra_Cell12, v_CellInfo12);

    //Configure SIB5 for Cell 12 according to Table 6.1.2.7.3.3-6

    //CellReselectionPriority for Cell 1 not present

    f_EUTRA_CellInfo_SetSIB5_6_1_2_7 ( eutra_Cell12, cs_508_SystemInformationBlockType5_Def ( { cs_InterFreqCarrierFreq_CRPri ( v_CarrierFreq_Cell1,

                                                                                         v_ChannelBandwidthDependency.AllowedMeasBandwidth,

                                                                                         omit ),

                                                                                       cs_InterFreqCarrierFreq ( v_CarrierFreq_Cell13,

                                                                                         v_ChannelBandwidthDependency.AllowedMeasBandwidth ) } ) );

    v_CellInfo13 := f_EUTRA_CellInfo_InitSIB5_6_1_2_7 (eutra_Cell13,

                                                     px_ePrimaryFrequencyBand,

                                                     c3,

                                                     px_ePrimaryBandChannelBandwidth,

                                                     v_UL_Bandwidth);  

    f_EUTRA_CellInfo_Set (eutra_Cell13, v_CellInfo13);    

    //Configure SIB5 for Cell 13 according to Table 6.1.2.7.3.3-7

    //CellReselectionPriority for Cell 1 not present

    f_EUTRA_CellInfo_SetSIB5_6_1_2_7 ( eutra_Cell13, cs_508_SystemInformationBlockType5_Def ( { cs_InterFreqCarrierFreq_CRPri ( v_CarrierFreq_Cell1,

                                                                                         v_ChannelBandwidthDependency.AllowedMeasBandwidth,

                                                                                         omit ),

                                                                                       cs_InterFreqCarrierFreq ( v_CarrierFreq_Cell12,

                                                                                         v_ChannelBandwidthDependency.AllowedMeasBandwidth ) } ) );
    //Set maximum cell powel level for Cell 1 & 12 to be used in creation

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell1, -67 );

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell12, -82 );




New Function:
	  function f_EUTRA_CellInfo_InitSIB5_6_1_2_7( CellId_Type        p_CellId,

                                  FrequencyBand_Type p_FrequencyBand,

                                  Combination_Type   p_SysinfoCombination,

                                  Dl_Bandwidth_Type  p_DL_Bandwidth,

                                  Ul_Bandwidth_Type  p_UL_Bandwidth ) return EUTRA_CellInfo_Type

  {

    var CellIdentities_Type v_CellIds  := fl_EUTRA_InitCellIdentities ( p_CellId );

    var EUTRA_FDD_TDD_CellInfo_Type v_FDD_TDD_Info := fl_InitFDD_TDD_FromBand (p_FrequencyBand);

    var EUTRA_FDD_TDD_Mode_Type v_FDD_TDD := v_FDD_TDD_Info.FDD_TDD;

    var CellTimingInfo_Type  v_Celltiming := fl_EUTRA_InitialiseCellTiming (p_CellId, v_FDD_TDD, LTE_AllCellsOnSamePLMN);   //the parameter "LTE_AllCellsOnSamePLMN" is dummy for RRC cells, it is used only for NAS test cases

    //var Combination_Type v_SysinfoCombination := fl_CheckSysinfoCombination (p_CellId, p_SysinfoCombination);

    var RootSequenceIndex_Type v_RootSequenceIndex_Type := fl_InitialisePRACH_RootSequenceIndex (p_CellId, v_FDD_TDD, LTE_AllCellsOnSamePLMN);

    var RACH_TimingAdvance_Type  v_RAR_TA := fl_InitialiseRAR_TA (p_CellId, v_FDD_TDD, LTE_AllCellsOnSamePLMN);

    var CarrierFreqEUTRA v_UL_DL_Earfcn;

    var NAS_Parameter_Type v_NAS_Param;

    var template (value) BcchInfo_Type  v_BCCH_Info;

    var template (value) EUTRA_CellInfo_Type v_CellInfo;

    var Frequency_fList_Type v_Frequency_PrimaryBand;

    var Frequency_fList_Type v_Frequency_SecondaryBand;

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam ( p_DL_Bandwidth, p_UL_Bandwidth );  //@sic R5s090180: return CFI value sic@

    var EUTRA_SysinfoDL_Bandwidth_Type v_SysinfoDL_Bandwidth := fl_ConvertDL_Bandwidth_ToSysinfoType (p_DL_Bandwidth);

    var template (value) BCCH_BCH_Message v_MIB := cs_508_MasterInformationBlock_Init ( v_SysinfoDL_Bandwidth );

    var template (value) BCCH_DL_SCH_Message v_SIB1;

    var template (value) SystemInformation_r8_IEs v_SIB2;

    var template (value) SystemInformation_r8_IEs v_SIB3;

    var template (value) SystemInformation_r8_IEs v_SIB5;

    var template (value) SystemInformationBlockType5 v_SystemInformation5;

    // Initialise Frequencies f1/f2/f3/f4

    v_Frequency_PrimaryBand := f_EUTRA_InitFrequency_f1Tof4 ( p_FrequencyBand, p_DL_Bandwidth, p_UL_Bandwidth );

    v_Frequency_SecondaryBand := f_EUTRA_InitFrequency_f1Tof4 (px_eSecondaryFrequencyBand, p_DL_Bandwidth, p_UL_Bandwidth);

    // Initialised Frequency per cell. See 36.508 Table 6.2.3.1-1

    // Cell 1/2/4/11 -> Frequency f1

    // Cell 3/12 -> Frequency f2

    // Cell 6/13 -> Frequency f3

    // Cell 14 -> Frequency f4

    select (p_CellId) {

      case (eutra_Cell12) {

        v_UL_DL_Earfcn := v_Frequency_PrimaryBand.f2;  // f2

      }

      case (eutra_Cell13) {

        v_UL_DL_Earfcn := v_Frequency_PrimaryBand.f3;   // f3

      }

      case else {

        FatalError (__FILE__, __LINE__, "invalid cell Idendtity");

      }

    }

    // Initialise NAS parameters

    v_NAS_Param := fl_EUTRA_InitNAS_forRRC_Cells( p_CellId, p_FrequencyBand );

    // Initialise SYSINFO

    v_SIB2 := cs_SI_SIB2 (cs_508_SystemInformationBlockType2_Def (omit,     // UL Bandwidth is omitted

                                                                    v_ChannelBandwidthDependency.N_RB_CQI,

                                                                    v_ChannelBandwidthDependency.Pusch_HoppingOffset,

                                                                    cs_508_SoundingRS_UL_ConfigCommon_FDD(v_ChannelBandwidthDependency.SrsBandwidthConfiguration),

                                                                    cs_508_PRACH_ConfigSIB_FDD ( v_ChannelBandwidthDependency.Prach_FrequencyOffset,

                                                                                                 v_RootSequenceIndex_Type ),  // @sic R5s090180: Replace _TDD by _FDD sic@

                                                                    tsc_P_b_1Tx

                                                                    ) ) ;

    v_SIB3 := cs_SI_SIB3 (cs_508_SystemInformationBlockType3_Def);

    select (p_CellId) {

        case (eutra_Cell12) {

             v_SystemInformation5 := cs_508_SystemInformationBlockType5_Def ({ cs_InterFreqCarrierFreq ( v_Frequency_PrimaryBand.f1, v_ChannelBandwidthDependency.AllowedMeasBandwidth ),

                                                              cs_InterFreqCarrierFreq ( v_Frequency_PrimaryBand.f3, v_ChannelBandwidthDependency.AllowedMeasBandwidth ),

                                                              cs_InterFreqCarrierFreq ( v_Frequency_SecondaryBand.f1, v_ChannelBandwidthDependency.AllowedMeasBandwidth ) });

        }

        case (eutra_Cell13) {

            v_SystemInformation5 := cs_508_SystemInformationBlockType5_Def ({ cs_InterFreqCarrierFreq ( v_Frequency_PrimaryBand.f1, v_ChannelBandwidthDependency.AllowedMeasBandwidth ),

                                                            cs_InterFreqCarrierFreq ( v_Frequency_PrimaryBand.f2, v_ChannelBandwidthDependency.AllowedMeasBandwidth ),

                                                            cs_InterFreqCarrierFreq ( v_Frequency_SecondaryBand.f1, v_ChannelBandwidthDependency.AllowedMeasBandwidth ) });

        }

    }

    v_SIB5 := cs_SI_SIB5 ( v_SystemInformation5 );

    v_SIB1 := cs_508_SystemInformationBlockType1_Def( cs_PlmnIdList_1Entry(v_NAS_Param.Guti_Parameters.PLMN_Identity),

                                                      v_NAS_Param.TrackingAreaCode,

                                                          v_CellIds.CellIdentity,

                                                          p_FrequencyBand,

                                                          cs_SchedulingInfo_3SI (v_ChannelBandwidthDependency.SIB3_periodicity,

                                                                                 {sibType3},

                                                                                 v_ChannelBandwidthDependency.SIB5_periodicity,

                                                                                 {sibType5} ),

                                                          omit );

    v_BCCH_Info := cs_BCCHInfo_C1( v_MIB,

                                          v_SIB1,

                                          { cs_SystemInformation_Common( v_SIB2 ),

                                            cs_SystemInformation_Common( v_SIB3 ),

                                            cs_SystemInformation_Common( v_SIB5 ) }

                                          );

    v_CellInfo := cs_EUTRA_CellInfoDef ( v_CellIds,

                                         p_SysinfoCombination,

                                         p_FrequencyBand,

                                         v_UL_DL_Earfcn,

                                         p_DL_Bandwidth,

                                         p_UL_Bandwidth,

                                         v_NAS_Param,

                                         v_BCCH_Info,

                                         v_Celltiming,

                                         v_FDD_TDD_Info,

                                         v_ChannelBandwidthDependency.Cfi,

                                         v_RAR_TA,

                                         v_ChannelBandwidthDependency.sr_PUCCH_ResourceIndex   // @sic R5s090312 sic@

                                         ); // @sic R5s090180 sic@

    return valueof(v_CellInfo);   // Note: valueof cannot be avoided here

  };


	    function f_EUTRA_CellInfo_SetSIB5_6_1_2_7 ( CellId_Type p_CellId,

                                      template (value) SystemInformationBlockType5 p_SIB5 ) runs on EUTRA_PTC

  {

    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get ( p_CellId );

    v_CellInfo.Sysinfo.BCCH_Info.SIs[2].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib5 := p_SIB5;

    f_EUTRA_CellInfo_Set (p_CellId, v_CellInfo);

  }


4.3 Change 2
	Testcase name
	f_TC_6_1_2_7_EUTRA

	Reason for change
	The guard timer is set to 420 seconds which is not sufficient for this test case to get to completion since idle mode generic timer which is of 360 seconds duration is run more than once during test case execution

	Summary of change
	Increased the guard timer duration to 600 seconds for this testcase.

	Source of change
	LTE_EPS_TS_Testcases_10wk33.ttcn


Before:

	  testcase TC_6_1_2_7() runs on MTC system SYSTEM {

    // @purpose

    //   Cell reselection / Equivalent PLMN

    var EUTRA_PTC   v_EUTRA     := EUTRA_PTC.create alive;

    var UTRAN_PTC   v_UTRAN     := null;

    var GERAN_PTC   v_GERAN     := null;

    var CDMA2000_PTC v_CDMA2000  := null;

    timer t_GuardTimer := int2float(420);

    f_MTC_ConnectPTCs(v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000);

    v_EUTRA.start   (f_TC_6_1_2_7_EUTRA());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }


After:

	  testcase TC_6_1_2_7() runs on MTC system SYSTEM {

    // @purpose

    //   Cell reselection / Equivalent PLMN

    var EUTRA_PTC   v_EUTRA     := EUTRA_PTC.create alive;

    var UTRAN_PTC   v_UTRAN     := null;

    var GERAN_PTC   v_GERAN     := null;

    var CDMA2000_PTC v_CDMA2000  := null;

    timer t_GuardTimer := int2float(600);
    f_MTC_ConnectPTCs(v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000);

    v_EUTRA.start   (f_TC_6_1_2_7_EUTRA());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }


5 Execution Log Files

5.1 Huawei CP V100 UE
The Huawai UE passed these test cases on the Anite Conformance Toolset solution with AES integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files tc_6_1_2_7-log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6 References

	[1]
	R5s100528: This archive comprises text format execution log file and the TTCN file.

